

Implicit Differentiation

Douglas Meade, Ronda Sanders, and Xian Wu

Department of Mathematics

Overview

This lab provides experience working with functions defined implicitly.

Maple Essentials

- The new Maple commands introduced in this lab are:

Command	Description	Example
<code>display</code>	combine one or more plots in a single plot; part of the <code>plots</code> package	<code>display([P1,P2], title="My Graph");</code>
<code>implicitdiff</code>	compute derivatives for implicitly-defined functions	Finding $\frac{dy}{dx}$: <code>implicitdiff(eq, y, x);</code> Finding $\frac{d^n y}{dx^n}$: <code>implicitdiff(eq, y, x\$n);</code>
<code>implicitplot</code>	create graph of function defined implicitly; part of the <code>plots</code> package	<code>implicitplot(eq, x=a..b, y=c..d);</code>
<code>pointplot</code>	plots a single point; part of the <code>plots</code> package	<code>pointplot([a,b], symbolsize=15);</code>
<code>fsolve</code>	compute a solution of equations numerically	<code>fsolve({eq1,eq2}, {x,y});</code>
<code>with</code>	loads the contents of a Maple package	<code>with(plots):</code>

- The *ImplicitDifferentiation* maplet is available from the course website:

<http://www.math.sc.edu/calclab/141L-S09/labs/> → [ImplicitDifferentiation](#)

Preparation

Review §4.1 Implicit Differentiation (Pages 235-247) in Anton.

Assignment

Exercises 18, 26, and 39 on pages 241-242.

Note: For part (c) of Exercise 39, you need to specify different regions in `fsolve` according to the graph to get all solutions.

Activities

- Find the equation of the tangent line to the curve $2(x^2 + y^2)^2 = 25(x^2 - y^2)$ at the point (3, 1). Then graph the curve, the point, and the tangent line together on one plot with a viewing window of $[-5, 5] \times [-4, 4]$. (Ex. 31 on page 242)
- Find all points where the tangent line to the graph of $x^2y - xy^2 = 2$ is horizontal or vertical.
- Find $\frac{d^2y}{dx^2}$ and $\frac{d^3y}{dx^3}$ if y is defined implicitly by $y + \sin y = x$. (Ex. 25 on page 242)

Example Problem

We will solve Example 5 on page 239 together using Maple:

- Use implicit differentiation to find $\frac{dy}{dx}$ for the Folium of Descartes $x^3 + y^3 = 3xy$.
- Find an equation of the tangent line to the Folium of Descartes at the point $(\frac{3}{2}, \frac{3}{2})$.
(Then graph the curve, the point, and the tangent line with a viewing window of $[-3, 3] \times [-4, 3]$ as shown in Figure 4.1.5 on page 239.)
- At what point(s) in the first quadrant is the tangent line to the Folium of Descartes horizontal?
(At what points is the tangent line vertical?)

Steps:

1. First, load the Maple `plots` package. Without the contents of this package, much of what we do today will not work.
> `with(plots):`
2. Assign our equation using `:=`.
> `eq:= x^3 + y^3 = 3*x*y;`
3. Find (and assign) the derivative using implicit differentiation. Since we want $\frac{dy}{dx}$, we input `y` and then `x`.
> `dydx:= implicitdiff(eq, y, x);`
4. Find (and assign) the slope of the tangent line at the point $(-1,1)$.
> `m:= eval(dydx, {x=3/2, y=3/2});`
5. Find (and assign) the equation of the tangent line. Remember: $y = m(x - x_1) + y_1$.
> `L:= m*(x - 3/2) + 3/2;`
6. Next, write (and assign) commands to plot the curve, the point, and the tangent line. Write the commands separately using `:` so Maple does not display the output yet. (In the first plot command, the option `numpoints=10000` will insure a smooth curve.)
> `P1:= implicitplot(eq, x=-3..3, y=-4..3, numpoints=10000):`
> `P2:= pointplot([3/2,3/2], color=green, symbolsize=15):`
> `P3:= plot(L, x=-3..3, y=-4..3, color=blue, linestyle=dash):`
7. Use the `display` command to display the curve, point, and tangent line on a single plot.
> `display([P1, P2, P3], title='Figure 1');`
8. From the graph, we can see that the tangent line would be horizontal at a point located approximately at $(1.2, 1.5)$. To find the point exactly, we need to find a point on the curve where $\frac{dy}{dx} = 0$. We can find this point using `fsolve`.
> `fsolve({eq, dydx=0}, {x,y}, {x=1..2, y=1..2});`
9. From the graph, we can see that the tangent line would be vertical at a point located approximately at $(1.5, 1.2)$. To find the point exactly, we need to find a point on the curve where $\frac{dy}{dx}$ is undefined. That is, a point where the denominator of $\frac{dy}{dx}$ is 0 We can find this point using `fsolve`.
> `fsolve({eq, denom(dydx)=0}, {x,y}, {x=1..2, y=1..2});`

Additional Notes

The `ImplicitDifferentiation` maplet provides additional practice finding the slope of a curve at a point.